# $Id: PKGBUILD 160504 2012-06-02 10:26:55Z bluewind $
# Maintainer: Kevin Piche <kevin@archlinux.org>
# Contributor: K. Piche <kpiche@rogers.com>

pkgname=perl-bit-vector
_cpanname=Bit-Vector
pkgver=7.2
pkgrel=2
pkgdesc='Efficient bit vector, set of integers and "big int" math library'
arch=('i686' 'x86_64')
license=('GPL' 'LGPL' 'PerlArtistic')
url="http://search.cpan.org/dist/${_cpanname}/"
depends=('perl-carp-clan' 'perl')
options=('!emptydirs')
source=(http://search.cpan.org/CPAN/authors/id/A/AN/ANDK/${_cpanname}-${pkgver}.tar.gz)
md5sums=('bc6ae34738b8f889290bf50099ff8aed')

build() {
  cd "${srcdir}/${_cpanname}-$pkgver"
  # install module in vendor directories.
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd "${srcdir}/${_cpanname}-$pkgver"
  make test
}

package() {
  cd "${srcdir}/${_cpanname}-$pkgver"
  make DESTDIR="${pkgdir}" install
}
